

# *Skew Densities and Ensemble Inference for Financial Economics*

**H. D. Vinod**

Financial economics includes the study of the density of ‘excess returns,’ having skewness, fat tails, and excess kurtosis. We illustrate it using monthly data on a mutual fund. The nonnormality of  $f(x)$  impacts measurement of potential losses by ‘value at risk’ (VaR). Since finance literature has mostly ignored Azzalini’s [1] skew normal (SN) density, our aim is to illustrate its implementation using *Mathematica* and *mathStatica* [2]. We use a location-scale version of the SN distribution, create its likelihood function, and use maximum likelihood (ML) to estimate its parameters. We illustrate it with data for a mutual fund and report the ML estimates, quantiles of  $f(x)$ , and the VaR.

## ■ 1. Introduction

Excess return from an asset (e.g., a mutual fund) is defined as the return over and above the risk-free rate (e.g., interest rate earned on the three-month U.S. Treasury Bills). Recent literature in financial economics recognizes that the density of excess returns  $f(x)$  is often nonnormal due to skewness, kurtosis, and fat tails. While some extensions of the normal called log-normal and inverse Gaussian have been used, Azzalini’s [1] SN density appears to have been ignored. The aim of this article is to introduce *mathStatica* implementation from [3]. This software also enables easy estimation of several other nonnormal  $f(x)$  for myriad potential uses in finance. We discuss ML estimators for the parameters of the SN density.

Wall Street investors, bankers, and government regulators often want a dollar figure on the potential loss in a worst-case scenario. VaR is one such measure developed by statisticians at RiskMetrics Group, a private company in New York. VaR is often described as based on historical volatility and tells the investor what might happen if some unusual event made the asset more volatile than normal. Although actual measurements differ among analysts, VaR is often obtained from a low (e.g., 1%) quantile of a parametric or nonparametric  $f(x)$ . Since mutual fund investors have seen three consecutive down years, both investors and regulators are concerned about fund VaR values. Morgenstern [4] notes that

risky funds in the aggressive growth stock category have recently experienced \$5.41 billion in redemptions compared to May 2002, whereas fund managers have increased assets in the aggressive category from \$430 billion to \$1.3 trillion. Our point is that VaR is important in finance, and we propose simple tools for calculating it with the help of *mathStatica*.

Section 2 considers the Alliance All-Asia Investment Advisors Fund mutual fund, with the ticker symbol AAAYX, for the period of  $T = 132$  months from January 1987 to December 1997 from [5]. Descriptive statistics reveal that the density  $f(x)$  is negatively skewed. Section 3 provides a generalised SN model adjusted for location and scale. Section 4 discusses ML estimation of the parameters of the generalised SN density. Section 5 briefly discusses VaR.

In future work, we expect to develop *mathStatica* software for a new maximum entropy algorithm (ME-*alg*) for the time series inference suggested in [6] for inference regarding the VaR. Classical time series inference methods treat the observed time series  $x_t$  as one random realization from an unobservable ensemble  $\Omega$ . These were developed in the 1930s before computers and before Shannon information and entropy ideas were developed. The ME-*alg* suggests a computer intensive construction of an approximate  $\Omega$  built up from elements  $\Omega_j$  for  $j = 1, 2, \dots, \mathcal{J}$  with  $\mathcal{J}$  a large number. Section 6 contains our conclusions.

We begin by loading the *mathStatica* package.

```
In[1]:= << mathStatica.m
```

## ■ 2. The Empirical Data

The data for AAAYX, *ydata*, is loaded as follows.

```
In[2]:= ydata = {10.724', 1.737', 2.123', 0.031', -0.483', 3.913', 4.284',
  2.275', -2.462', -21.147', -6.062', 6.318', 3.221', 2.623',
  -3.985', 0.792', 1.644', 4.042', -1.892', -3.922', 2.683', 0.673',
  -1.167', 1.11', 5.084', -3.543', 0.24', 3.783', 3.165', -2.281',
  5.836', 2.028', -1.974', -3.262', 2.253', 2.256', -6.66', 2.338',
  0.935', -4.164', 6.84', -2.66', -0.538', -9.629', -4.608',
  -0.705', 5.103', 2.153', 3.574', 4.487', -0.388', -0.568', 3.93',
  -4.971', 2.93', 1.948', -0.134', 1.391', -4.869', 8.073', -1.325',
  1.167', -1.343', 3.987', 0.887', -1.302', 3.231', -2.667',
  1.573', -0.45', 2.824', 2.04', -0.145', 2.051', 2.756', -0.746',
  0.741', 0.144', 0.644', 3.246', -0.447', 0.444', -1.266', 2.146',
  2.55', -2.184', -4.595', 0.572', 1.743', -1.852', 2.738', 4.213',
  -1.996', 0.973', -4.072', 2.03', 0.505', 4.011', 2.116', 1.517',
  3.122', 1.038', 1.84', 0.651', 2.955', -2.352', 5.648', 1.379',
  2.483', 0.585', 0.976', 0.275', 0.772', -1.428', -4.54', 2.863',
  4.484', 2.774', 6.676', -2.028', 3.673', 1.067', -2.84', 3.661',
  3.971', 3.871', 7.067', -5.332', 4.383', -4.93', 3.762', 0.058'};
```

The descriptive statistics for our data are given by using `Statistics`DescriptiveStatistics``.

```

In[3]:= << Statistics`DescriptiveStatistics`
In[4]:= LocationReport[ydata]
Out[4]= {Mean → 0.738705, HarmonicMean → 1.95493, Median → 1.0885}
In[5]:= DispersionReport[ydata]
Out[5]= {Variance → 14.1743, StandardDeviation → 3.76488,
  SampleRange → 31.871, MeanDeviation → 2.68482,
  MedianDeviation → 2.055, QuartileDeviation → 2.11325}
In[6]:= ShapeReport[ydata]
Out[6]= {Skewness → -1.58578,
  QuartileSkewness → -0.122678, KurtosisExcess → 7.69132}

```

These reports show that the usual normal distribution is not suitable for our data and that the negative skewness requires extra care in interpreting the results. To illustrate graphically, *mathStatica* can be used for nonparametric kernel density estimation in two steps. First, we specify the kernel as the Gaussian kernel. The second step is to choose the bandwidth denoted here by  $c$ .


```

In[7]:= Gauss =  $\frac{e^{-\frac{y^2}{2}}}{\sqrt{2\pi}}$ ;
  domain[Gauss] = {y, -∞, ∞};
In[9]:= c = Bandwidth[ydata, Gauss, Method → SheatherJones]
Out[9]= 1.104

```

Since the kernel  $K$  and  $c$  have now been specified, we can plot Figure 1 as our smoothed nonparametric kernel density estimate using the `NPKDEPlot[data, K, c]` function.

```
In[10]:= NPKDEPlot[ydata, Gauss, c];
```



**Figure 1.**

The preceding discussion and plot illustrate a common situation: normality is often not suitable for many financial time series including ours. *mathStatica* can

effectively implement a wide range of nonnormal densities for applications in finance.

### ■ 3. Model: A Generalised Azzalini Skew Normal Distribution

Next, we illustrate the power of *mathStatica* and *Mathematica* in financial economics by implementing a generalised version of Azzalini’s SN density in the context of our example. Azzalini [1] proved that if a random variable  $X$  has a pdf  $f(x)$  which is symmetric about 0, and cdf  $F(x)$ , then  $g(x) = 2 f(x) F(\lambda x)$  is also a pdf, for parameter  $\lambda \in \mathbb{R}$ . If  $X \sim N(0, 1)$ , then  $g(x)$  is Azzalini’s SN density. Unfortunately, the latter does not contain sufficient flexibility for our financial application which will require both a scale parameter (for large variance) and a location parameter.

To start our generalisation, we first introduce a scale parameter  $w > 0 \dots$  hence, let  $X \sim N(0, w^2)$  with pdf  $f(x)$ .

$$\text{In[11]:= } \mathbf{f} = \frac{1}{w \sqrt{2 \pi}} \text{Exp}\left[-\frac{x^2}{2 w^2}\right];$$

$$\text{domain}[\mathbf{f}] = \{\mathbf{x}, -\infty, \infty\} \&\& \{\mathbf{w} > 0\};$$

The cdf evaluated at  $F(\lambda x)$  is:

$$\text{In[13]:= } \mathbf{F} = \text{Prob}[\lambda \mathbf{x}, \mathbf{f}]$$

$$\text{Out[13]= } \frac{1}{2} \left( 1 + \text{Erf}\left[\frac{x \lambda}{\sqrt{2} w}\right] \right)$$

Then, our generalised Azzalini skew  $N(0, w^2)$  density,  $g(x)$ , is simply:

$$\text{In[14]:= } \mathbf{g} = 2 * \mathbf{f} * \mathbf{F}$$

$$\text{Out[14]= } \frac{e^{-\frac{x^2}{2 w^2}} \left( 1 + \text{Erf}\left[\frac{x \lambda}{\sqrt{2} w}\right] \right)}{\sqrt{2 \pi} w}$$

with domain of support:

$$\text{In[15]:= } \text{domain}[\mathbf{g}] = \{\mathbf{x}, -\infty, \infty\} \&\& \{\lambda \in \text{Reals}, \mathbf{w} > 0\};$$

Next, we introduce a location parameter  $\xi$  into the model by transforming  $X \rightarrow Y$  such that  $Y = X + \xi$ . This is facilitated by *mathStatica*’s `Transform[ ]` function to yield pdf  $h(y)$ .

$$\text{In[16]:= } \mathbf{h} = \text{Transform}[\mathbf{y} = \mathbf{x} + \xi, \mathbf{g}]$$

$$\text{Out[16]= } \frac{e^{-\frac{(y-\xi)^2}{2 w^2}} \left( 1 + \text{Erf}\left[\frac{\lambda (y-\xi)}{\sqrt{2} w}\right] \right)}{\sqrt{2 \pi} w}$$

with domain of support:

$$\text{In[17]:= } \text{domain}[\mathbf{h}] = \{\mathbf{y}, -\infty, \infty\} \&\& \{\lambda \in \text{Reals}, \xi \in \text{Reals}, \mathbf{w} > 0\};$$

Figure 2 plots our generalised location-scale SN density for three values of the key parameter  $\lambda$  to verify that its negative values lead to negative skewness in a graph. Note that if a mutual fund has  $\lambda = -9$  (other things being equal), it is mostly losing money. In the sequel, we shall estimate  $\lambda$  by maximizing the likelihood function.

```
In[18]:= PlotDensity[h /. {ξ → 0, w → 1, λ → {0, -2, -9}}, {y, -3, 3}];
```




Figure 2.

## ■ 4. Maximum Likelihood Estimation

Next, we activate *mathStatistica*'s `SuperLog` function for subsequent calculation of the log likelihood for  $b(y)$ .

```
In[19]:= SuperLog[On]
```

SuperLog is now On.

Now we are ready to define  $\log L\theta$ , the log of the likelihood function, from a product of  $b(y)$  densities over  $n$  observations.

```
In[20]:= logLθ = Log[∏_{i=1}^n (h /. y → y_i)]
```

$$\text{Out[20]} = -\frac{1}{2w^2} \left( n\xi^2 + n w^2 \text{Log}[2] + n w^2 \text{Log}[\pi] + 2n w^2 \text{Log}[w] - \right. \\ \left. 2w^2 \sum_{i=1}^n \text{Log}\left[1 + \text{Erf}\left[\frac{\lambda(-\xi + y_i)}{\sqrt{2}w}\right]\right] - 2\xi \sum_{i=1}^n y_i + \sum_{i=1}^n y_i^2 \right)$$

The ML estimates of the three parameters  $\xi$ ,  $\lambda$ , and  $w$  require differentiation of the  $\log L\theta$ , leading to three so-called score functions.

In[21]:= **score = Grad[logLθ, {ξ, λ, w}]**

$$\text{Out[21]} = \left\{ -\frac{2n\xi - 2w^2 \sum_{i=1}^n -\frac{e^{-\frac{\lambda^2(-\xi+y_i)^2}{2w^2}} \sqrt{\frac{2}{\pi}} \lambda}{w(1+\text{Erf}\left[\frac{\lambda(-\xi+y_i)}{\sqrt{2}w}\right])} - 2 \sum_{i=1}^n y_i}{2w^2}, \right.$$

$$\sum_{i=1}^n \frac{e^{-\frac{\lambda^2(-\xi+y_i)^2}{2w^2}} \sqrt{\frac{2}{\pi}} (-\xi+y_i)}{w(1+\text{Erf}\left[\frac{\lambda(-\xi+y_i)}{\sqrt{2}w}\right])},$$

$$\frac{1}{w^3} \left( n\xi^2 + n w^2 \text{Log}[2] + n w^2 \text{Log}[\pi] + 2n w^2 \text{Log}[w] - \right.$$

$$2w^2 \sum_{i=1}^n \text{Log}\left[1 + \text{Erf}\left[\frac{\lambda(-\xi+y_i)}{\sqrt{2}w}\right]\right] - 2\xi \sum_{i=1}^n y_i + \sum_{i=1}^n y_i^2 \left. \right) -$$

$$\frac{1}{2w^2} \left( 2nw + 2nw \text{Log}[2] + 2nw \text{Log}[\pi] + 4nw \text{Log}[w] - \right.$$

$$4w \sum_{i=1}^n \text{Log}\left[1 + \text{Erf}\left[\frac{\lambda(-\xi+y_i)}{\sqrt{2}w}\right]\right] -$$

$$\left. 2w^2 \sum_{i=1}^n -\frac{e^{-\frac{\lambda^2(-\xi+y_i)^2}{2w^2}} \sqrt{\frac{2}{\pi}} \lambda (-\xi+y_i)}{w^2(1+\text{Erf}\left[\frac{\lambda(-\xi+y_i)}{\sqrt{2}w}\right])} \right\}$$

These equations are too complicated to analytically obtain the ML estimators by setting the scores equal to zero. We have to use numerical solution methods on our data.

The observed log likelihood is obtained by simply inserting the data values in logLθ.

In[22]:= **obslogLθ = logLθ /. {n → Length[ydata], y\_i\_ → ydata[[i]]};**

We then use the FindMaximum function to numerically maximize the observed log likelihood, as follows:

In[23]:= **sol = FindMaximum[obslogLθ, {{ξ, 0.1}, {λ, -0.2}, {w, 2}}]**

FindMaximum::lstol :  
 The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient increase in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances. [More...](#)

Out[23]= {-361.793, {ξ → 0.738673, λ → 0.0000106647, w → 3.75059}}

In the preceding output, the first component of the solution is the value of the maximized likelihood. The second component is sol[[2]] and represents the ML solution for the three parameters. Mathematica’s warning message suggests that

we should try other methods and starting points. We will try another search, now using `Method→Gradient` and slightly different starting points.

```
In[24]:= sol1 = FindMaximum[obslogLθ, {ξ, 0.78},
      {λ, -0.014}, {w, 3.75}, Method → Gradient]
```

```
Out[24]= {-361.342, {ξ → 2.40441, λ → -0.55144, w → 3.779}}
```

The new output in `sol1` is already an improvement. But is it the maximum? By the first-order conditions, the gradients of the observed  $\log L\theta$  evaluated at `sol1[[2]]` should be zero.

```
In[25]:= grd = Grad[obslogLθ, {ξ, λ, w}];
      grd /. sol1[[2]]
```

```
Out[26]= {-2.18193, -6.30854, 5.34271}
```

These values of the gradient clearly are not close enough to zero. Similarly, the second-order conditions for a maximum require that the matrix  $H$  of second-order partials be negative definite. `Hessian` calculates this matrix, and `Eigenvalues` calculates its eigenvalues as follows:

```
In[27]:= hess = Hessian[obslogLθ, {ξ, λ, w}]; Eigenvalues[hess /. sol1[[2]]]
```

```
Out[27]= {-61.5201, -25.7395, 0.293474}
```

Unfortunately, the last eigenvalue, although small, does not ensure negative definiteness. The issue is whether the observed log likelihood is concave near the solution. McCullough and Vinod [7, 8] recommend searching near all ML solutions to be certain. Fortunately, *Mathematica* makes these searches relatively easy to implement by specifying an alternative method. Hence we try the Newton estimation method.

```
In[28]:= sol2 = FindMaximum[obslogLθ,
      {ξ, 0.78}, {λ, -0.014}, {w, 3.75}, Method → Newton]
```

```
Out[28]= {-354.229, {ξ → 4.54772, λ → -2.30187, w → 5.34561}}
```

Is this solution (denoted by `sol2`) superior to the first one? The maximized value of the log likelihood has increased from  $-361.342$  to  $-354.229$ . So the solution has clearly improved. The solution for the location parameter  $\xi$  is not comparable to the sample mean  $0.7387$  for our mutual fund, because of the scale change by  $w$ . The comparable value is obtained as:

```
In[29]:= ξ2 = ξ / w;
      ξ2 /. sol2[[2]]
```

```
Out[30]= 0.85074
```

To determine if `sol2` is good enough we study its properties as before.

```
In[31]:= grd /. sol2[[2]]
```

```
Out[31]= {-1.89938 × 10-8, -2.48643 × 10-8, 3.99092 × 10-9}
```

```
In[32]:= Eigenvalues[hess /. sol2[[2]]]
```

```
Out[32]= {-16.0411, -12.9347, -1.70855}
```

This solution has the desirable property that the gradients are very close to zero. The shape of the log likelihood near this second ML solution is concave, since the Hessian is negative definite. For a graphical assessment of the goodness-of-fit of the model to the data, we plot the data (solid line) and the fitted location-scale SN density  $b(y)$  (dashed line) side by side in Figure 3.

```
In[33]:= FrequencyPlot[ydata, h /. sol2[[2]]];
```




Figure 3.

Since this fit is visually close, we will accept `sol2` as our solution. The next step is to consider the statistical inference for the Newton solution of the three parameters. It is well known that the negative of the inverse of the Hessian matrix evaluated at the solution provides the asymptotic variance-covariance matrix of the ML parameter estimates.

```
In[34]:= asycov = -Inverse[hess /. sol2[[2]]]
```

$$\text{Out[34]} = \begin{pmatrix} 0.216946 & -0.180039 & 0.154585 \\ -0.180039 & 0.289606 & -0.174896 \\ 0.154585 & -0.174896 & 0.218391 \end{pmatrix}$$

The standard errors of the ML estimates are given by the diagonals of the preceding matrix. Hence we write student's t statistics as:

$$\text{In[35]} := \frac{\{\xi, \lambda, w\} /. \text{sol2}[[2]]}{\sqrt{\text{Table}[\text{asycov}[[i, i]], \{i, 3\}]}}$$

$$\text{Out[35]} = \{9.76378, -4.27738, 11.4388\}$$

Note that the t statistics are scale-free and suggest statistical significance of the three parameters at conventional confidence levels. We conclude that this mutual fund yields statistically significant positive return even after allowing for negative skewness in the data. The ML estimate of the skewness parameter,  $\lambda$  is  $-2.30187$ , is also of interest in financial economics in the following sense. Note that except for location and scale change, the density for our fund is close to the dotted line in Figure 3. If there are two funds with identical mean and variance but distinct skewnesses, then the investor would be better off choosing the fund with a larger (more positive) skewness.

The asymptotic inference may not be satisfactory in relatively small samples in finance. In future work we expect to develop a new maximum entropy algorithm (ME-*alg*) for the time series inference suggested in [6] for such inference.

## ■ 5. Value at Risk

The VaR is a popular measure of potential loss used by investors, bankers, and regulators. Let  $\Delta P(\tau) = P(t + \tau) - P(t)$  be the change in the value of a portfolio at time  $t$  for horizon  $\tau$  and let  $\alpha$  ( $< 0.5$ ) denote a probability. The VaR is defined by the probability statement:

$$\Pr[\Delta P(\tau) < -\text{VaR}] = 1 - \alpha, \quad (1)$$

where the negative VaR is designed to measure losses in positive dollars. Intuitively, VaR measures a worst-case scenario loss associated with 'long' positions (buying side). For example, consider an investor with a time horizon  $\tau$  of one year buying \$100,000 worth of mutual fund shares,  $P(t) = 100000$ . Now assume that the fund could lose 25% or more of its value in a year with probability  $\alpha = 0.01$ . Then  $\Pr[\Delta P(\tau) < 25000] = 0.99$ , implying that VaR = \$25,000 is an upper bound on the loss. Note that we can compute the VaR for any density whose cdf is known, at least numerically.

```
In[36]:= Quantile[ydata, 0.01]
```

```
Out[36]= -9.629
```

For our original data in terms of returns, the low return of  $-9.629\%$  suggests VaR = \$9629. It is customary not to use the estimate based on the empirical cdf, since the investor is concerned with worst-case scenarios, beyond what was experienced. One way to do this has been to use the lower one percentile of  $N(0, 1)$ , namely  $-2.33$ .

```
In[37]:= Mean[ydata] - 2.33 * Sqrt[Variance[ydata]]
```

```
Out[37]= -8.03347
```

This suggests that VaR = \$8033. It is somewhat surprising that the VaR from  $N(0, 1)$  is less than \$9629, from the quantile of the original data. We have not yet computed the VaR from  $h(y)$ . First, let us check its cdf as:

```
In[38]:= Prob[y, h]
```

$$\text{Out[38]} = \int_{-\infty}^y \frac{e^{-\frac{(y-\xi)^2}{2w^2}} (1 + \text{Erf}\left[\frac{\lambda(y-\xi)}{\sqrt{2}w}\right])}{\sqrt{2\pi}w} dy$$

Since inverting this analytically is not feasible, we use numerical methods as follows. A numerical version of the density is:

```
In[39]:= h2 = h /. sol2[[2]]
```

```
Out[39]= 0.0746299 e^{-0.0174975 (-4.54772+y)^2} (1 + Erf[-0.304487 (-4.54772 + y)])
```

The numerical cdf is obtained by integrating  $b^2(y)$  as follows:

```
In[40]:= NProb[r_?NumericQ] := NIntegrate[h2, {y, -∞, r}]
```

Next, we ask *Mathematica* to search for the value of  $Y$  at which the numerical cdf is equal to 0.01.

```
In[41]:= FindRoot[NProb[r] == 0.01, {r, -15, -5}]
```

```
Out[41]= {r → -9.22166}
```

Hence the VaR estimated from the location-scale SN density is \$9222, rounded to the nearest dollar. Vinod and Morey [9] note that financial economists usually ignore the estimation risk. Since this is also true in calculated VaR estimates, there is a need to develop inference methods for VaR. In future work, using tools described in Section 2.6 of [3] we seek to compute pseudo-random number generation from our  $b(y)$  and also from the kernel density plotted in Figure 1. The large number ( $\mathcal{J} = 999$ , say) of realizations of VaR can then be ordered from the smallest to the largest and denoted by  $\text{VaR}(j)$  with  $j = 1, 2, \dots, \mathcal{J}$ . An obvious possibility is to choose the  $\text{VaR}(10)$  as the estimate of VaR which incorporates both estimation risk and investment risk.

## ■ 6. Conclusion

We provide descriptive statistics and graphics which show that monthly excess returns data for a mutual fund reveals that their distribution  $f(x)$  is nonnormal. Since the SN distribution has been generally ignored by financial economists, we consider its potential in describing  $f(x)$ . Using *mathStatistica*, [2], we derive the score function for the parameters  $\xi$ ,  $\lambda$ , and  $w$  of a location-scale SN density and their ML estimators. We note that *mathStatistica* and *Mathematica* provide tools for numerous potential uses in finance, including ML estimation of parameters of several nonnormal  $f(x)$ . We show that these densities can be further used to compute low percentiles (1%) to help compute the VaR, which is gaining considerable use by Wall Street practitioners and the financial media [4].

## ■ References

- [1] A. Azzalini, "A Class of Distributions Which Includes the Normal Ones," *Scandinavian Journal of Statistics*, **12**, 1985 pp. 171–178.
- [2] *mathStatistica*, Sydney, Australia: MATHSTATICA Pty. Ltd., (Feb 2005) [www.mathStatistica.com](http://www.mathStatistica.com).
- [3] C. Rose and M. D. Smith, *Mathematical Statistics with Mathematica*, New York: Springer-Verlag, 2002.
- [4] G. Morgenstern, "How Big Could the Next Wave Be? Check the Risk Level," Market Watch section, *New York Times*, Feb 1, 2003.
- [5] Morningstar, *On-Disk, Principia, Principia-Plus, and Principia-Pro Manuals* (1991–2000), Chicago: Morningstar Publications, 2000.

- [6] H. D. Vinod, "Maximum Entropy Constructive Ensembles for Time Series Analysis" (discussion paper, Department of Economics, Fordham University, Bronx, NY, Dec 22, 2002). Paper presented at the 35th Symposium on the Interface: Computing Science and Statistics (Salt Lake City, UT, Mar 14, 2003) is available on CD-ROM from the Interface Foundation of North America, Inc., P. O. Box 7460, Fairfax Station, VA 22039-7460. Also available at [www.american.edu/cas/econ/faculty/golan/Papers/9\\_17vinod.doc](http://www.american.edu/cas/econ/faculty/golan/Papers/9_17vinod.doc).
- [7] B. D. McCullough and H. D. Vinod, "The Numerical Reliability of Econometric Software," *Journal of Economic Literature*, **37**(2), 1999 pp. 633–665.
- [8] B. D. McCullough and H. D. Vinod, "Verifying the Solution from a Nonlinear Solver: A Case Study," *The American Economic Review*, **93**(3), June 2003 pp. 873–892.
- [9] H. D. Vinod and M. R. Morey, "Estimation Risk in Morningstar Fund Ratings," *Journal of Investing*, **11**(4), 2002 pp. 67–75.

## About the Author

Professor H. D. Vinod is a professor of economics and director of the Institute for Ethics and Economic Policy at Fordham University in New York. He received his Ph.D. in economics from Harvard University, has worked at Bell Laboratories for many years, and has been an expert witness in major antitrust trials. He has published over a hundred articles in top journals in econometrics and statistics, including *Econometrica*, *Journal of Econometrics*, *Journal of the American Statistical Association*, *Management Science*, *American Economic Review*, *Review of Economics and Statistics*, *International Economic Review*, *Journal of Economic Literature*, *Journal of Business*, and *Journal of Advertising Research*. He has edited special issues and refereed many journals. In 1996, he was named a fellow of the *Journal of Econometrics*. Maharashtra Foundation honored him for outstanding social service in 1998. Vinod coauthored a 1981 research monograph on regression methods and a 2005 Wiley monograph, "Preparing for the worst: Incorporating downside risk in stock market investing." He co-edited volume 11 of the *Handbook of Statistics* with C. R. Rao, the famous statistician. Vinod manages a worldwide anticorruption project described at [www.fordham.edu/economics/vinod](http://www.fordham.edu/economics/vinod).

### H. D. Vinod

Professor  
Economics Department  
Fordham University  
Bronx, New York 10458  
[Vinod@fordham.edu](mailto:Vinod@fordham.edu)